
1 ロレンツモデル

山も一度登ってそれっきりというよりも再度登ったり、登りなれたコースを変えて山で風

景を愛ずるのもよい。

ロレンツの非線形連立方程式モデルを題材にかって使ったスクリプトを再度題材にして

みよう。

ここでは連立非線形差分方程式のスクリプトと連立非線形微分方程式の係数をマトリク

スに落とし込んで、ルンゲクッタ法で数値解を求める方法を取り上げる。

ロレンツモデルは非圧縮条件、ナビエ ·ストークス方程式、熱伝導方程式の 3式から導か
れる。

そしてこの 3個のパラメータは σがプランドル数、γがレイリー数、bが系の物理サイズ

に係わるものである。 
X
′
= −σX +σY

Y
′
= −XZ +γX −Y

Z
′
= XY −bZ

σ = 10, b =
3

8
, γ = 28がロレンツのオリジナル ·パラメータである。

書きためた論稿の中に 3種のロレンツの差分方程式のスクリプトがあった。

1. APL協会の山下、西川両氏の発表されたロレンツモデルの差分方程式とカオスモデル
には長年親しんできた。

2. 差分方程式のタシット型定義

3. マトリクスとルンゲクッタ-法による解法

1.1 山下、西川

ロレンツの方程式 (1963) を離散化し、漸加法で構成する山下, 西川によるスクリプトが
JAPLAで公表されている。

*1

このスクリプトはいきなり差分方程式を書き下ろしている。


dx

dt
= σ(y− x)

dy

dt
= x(ρ− z) − y

dz

dt
= xy− βz

σ is Prandtl number

ρ is Rayleigh number

σ, ρ, β > 0

usuallyσ = 10, ρ = 28, β = 8r3

dt=: 0.005 NB. or 0.01

%NB.10 50 8r3&calc_lorenz plot_lorenz init

lorenz1=:4 : 0

NB. Lorenz attracta

’s r b’=. x NB. parameter

’xx yy zz’=: y NB. initial 5 8 10 or 1 1 1

X=: xx + dt*s*(yy-xx)

Y=: yy + dt*((r*xx)-(yy+xx*zz))

Z=: zz + dt*((xx*yy)-b*zz)

X,Y,Z

)

’noaxes’ plot {@|: 10 28 8r3 lorenz1 ˆ:(i.2000) 0.1 0.2 0.3

pd ’save jpg c:/temp/lorenz_10.jpg’

図 1 Lorenz

*1 パラメーターと初期値を分離するため若干改良した。

1.2 もう一つのロレンツアトラクタのスクリプト

J の達人がタシット形式で書いたメモが残っていた。スクリプト自体が鑑賞の対象に
なる。

load ’plot’

’‘X Y Z’=: (0&{)‘(1&{)‘(2&{)

’s b r’=: 10 8r3 28

dx=: s * (Y - X)

dy=: (X * r - Z) - Y

dz=: (X*Y) - b*Z

dt=: 0.01

I=: + (dt * dx, dy, dz)

NB. Usage: ’noaxes’ plot <"1|: Iˆ:(i.5000) 0 1 0

’noaxes’ plot <"1|: Iˆ:(i.5000) 0 1 0

1.3 (1)と (2)の計算結果の比較

パラメータと初期値、dtを同じくすると lorenz1と I(lorenz2)の結果は同一である。

dt=:0.02

a1=: 10 28 8r3 lorenz1 ˆ:(i.10) 1 1 1

a2=. I ˆ:(i.10) 1 1 1

a1,._,.a2

NB. ---------------------------------------------------

1 1 1 _ 1 1 1

1 1.52 0.966667 _ 1 1.52 0.966667

1.104 2.03027 0.945511 _ 1.104 2.03027 0.945511

1.28925 2.58702 0.939912 _ 1.28925 2.58702 0.939912

1.54881 3.23303 0.95649 _ 1.54881 3.23303 0.95649

1.88565 4.00607 1.00562 _ 1.88565 4.00607 1.00562

2.30974 4.94399 1.10307 _ 2.30974 4.94399 1.10307

2.83659 6.08761 1.27263 _ 2.83659 6.08761 1.27263

3.48679 7.48215 1.55012 _ 3.48679 7.48215 1.55012

3



4.28586 9.17701 1.98922 _ 4.28586 9.17701 1.98922

(lorenz1) (lorenz2)

2 ロレンツモデルとルンゲクッタ法

2.1 C.Reiterの方法

C.Reiterの「Fractal visualization and J第 3版」でマトリクスとルンゲクッタ法でこの微
分方程式を計算しているのを見受けた。

1. C.Reiterによるロレンツのマトリクス。横長のマトリクスを用いて巧妙に作成されて
いる。
*2


X
′
= −σX +σY

Y
′
= γX −Y −XZ

Z
′
= XY −bZ

x y z xy xz

X ′ −10 10 0 0 0

Y ′ 28 −1 0 0 −1

Z ′ 0 0 −8r3 1 0

Y
′
= γX− Y −XZの −XZと Z

′
= XY − bZの XY を別項に書き出している。非線形

方程式も横長マトリクスで計算できる。
*3

M

_10 10 0 0 0

28 _1 0 0 _1

0 0 _8r3 1 0

2. 初期値の x, y, zに加え、xy, xzを作成している。（ここがポイント）

(] , {. * }.) 0.1 0.2 0.3

0.1 0.2 0.3 0.02 0.03 NB. make x y z xy xz

*2 かって線形化の好例として紹介したが、線形化を意識しないで、素直にマトリクスに乗せた方が良いようだ
*3 ネスティッド ·アレー（2重配列）や文字列操作などがなくとも表現できる

3. C.Reiterの lorenz3zの script

NB. Lorenz3

lorenz3z=: 1 : 0

’S B R’=. m NB. s b r

M=. ((-S),S,0 0 0),(R,_1 0 0 _1),: 0 0 ,(-B),1 0 NB. make matrix

M&(+/ . *)@(] , {. * }.) NB. 計算部分

)

（a） +/ . * は内積計算を行う。M (+/ . *) 初期値
（b）初期値の項数はマトリクスMの列数と等しくなければならない
（c）出力はMの行数すなわち連立方程式の元の数となる。
（d）ループ ˆ:(i. n) では出力を 1回づつ受けた上で xz, yzをその都度構成して入

力する

4. C.Reiterによるルンゲクッタ法のスクリプト。ループはタシットの外付けなので、実
にシンプルである。

ルンゲークッタ法はコンピューター以前の 1900年ごろに提唱されている。
差分計算はルンゲクッターに委ねて疑似的に行うことで更にスクリプトが簡約される。

rk=:1 : 0

:

h2=. -: x

k1=. u y

k2=. u y + h2*k1

k3=. u y + h2*k2

k4=. u y + x *k3

y+(x%6)*k1+k4++:k2+k3

)

• 1 : 0は動詞 (関数)を引数に取る副詞型
• uは左の動詞を引数入力
• +: double ×2

5. Usage:
0.002 (10 8r3 28 lorenz3) rk ˆ:(i.100) 0.1 0.2 0.3

’noaxes’ plot {@|: 0.002 (10 8r3 28 lz) rk ˆ:(i.2000) 0.1 0.2 0.3

6. run

2.2 (1)と (3)の比較

差分計算とルンゲクッタ法の比較。良い線を行っていると思われる。

a1=: 10 28 8r3 lorenz1 ˆ:(i.10) 0.1 0.2 0.3

a3=. 0.01 (10 8r3 28 lorenz3) rk ˆ:(i.10) 0.1 0.2 0.3

a1,._,.a3

NB. -------------------------------------

0.1 0.2 0.3 _ 0.1 0.2 0.3

0.11 0.2257 0.2922 _ 0.110802 0.22703 0.292328

0.12157 0.253922 0.284656 _ 0.123288 0.257007 0.284915

0.134805 0.285076 0.277374 _ 0.137599 0.290386 0.277769

0.149832 0.319597 0.270362 _ 0.153908 0.327662 0.270904

0.166809 0.357949 0.263631 _ 0.172424 0.369377 0.264336

0.185923 0.400636 0.257198 _ 0.193385 0.41613 0.258089

0.207394 0.44821 0.251084 _ 0.217072 0.468585 0.252193

0.231476 0.501277 0.245318 _ 0.243801 0.527484 0.246688

0.258456 0.56051 0.239937 _ 0.273937 0.593653 0.241628

lorenz1 lorenz3(runge-kutta)

2.3 ロスラーモデル

1976年にドイツの化学者ロスラーが提唱したロレンツモデルを簡略したカオスを生じる
モデル。


dx

dt
= −y− z

dy

dt
= x+ ay

dz

dt
= b+ (x+ c)z

パラメーターはロスラーは次とした。

a = b = 0.2 c = 5.7

次のもよく用いられる。

a = b = 0.1 c = 14


dx

dt
= −y− z

dy

dt
= x+ ay

dz

dt
= b+ (x+ c)z

x y z xz b

X ′ 0 −1 −1 0 0

Y ′ 1 a 0 0 0

Z ′ 0 0 −c 1 b

マトリクスをパラメーター a = b = 0.2 c = 5.7を用いて構成したもの。bは定数項
なので作成に少し工夫が要る

M2

0 _1 _1 0 0

1 0.2 0 0 0

0 0 _5.7 1 0.2

’noaxes’ plot {@|: 0.02 (0.2 0.2 5.7 rossler) rk ˆ:(i.2000) 0.1 0.2 0.3

NB. rossler

rossler=: 1 : 0

’a b c ’=. m NB. a b c

M2=. (0 _1 _1 0 0),(1,a,0 0 0),: 0 0 ,(-c),1 ,b NB. make matrix

M2&mp@mk_init2 NB. 計算部分

)

mk_init2=: ] , ({. * {:),1:

• mp=:+/ . * NB. 内積

• mk init2は初期値やリターン値を係数マトリクスの形に成形する
• 最後の 1:は常に 1を出力する動詞であり、定数項 (b)の個所を示している

